

Excel voor Gevorderden

Beginners kennen functies.

Gevorderden combineren ze alsof het Lego is.

Excel voor Gevorderden

Zelfverzekerd van basis naar gevorderd niveau
(Opvolger van "Excel voor Beginners")

Ad Kooij

Auteur: Ad Kooij

Coverdesign: Ad Kooij

Eindredactie: www.schrijfspa.nl

Editie: 1

ISBN: 978-94-038-2599-1

Website: <https://excelbooks7.wordpress.com>

© 2025 Ad Kooij

Niets uit deze uitgave mag worden verveelvoudigd, opgeslagen in een geautomatiseerd gegevensbestand of openbaar gemaakt, in enige vorm of op enige wijze, hetzij elektronisch, mechanisch, door fotokopieën, opnamen of op welke andere manier dan ook, zonder voorafgaande schriftelijke toestemming van de auteur.

Voor educatief gebruik op scholen of instellingen: neem contact op met de auteur.

Disclaimer:

Microsoft®, Excel® en andere merknamen zijn geregistreerde handelsmerken van Microsoft Corporation.

Dit boek is onafhankelijk geschreven en niet gelieerd aan of goedgekeurd door Microsoft.

Inhoud

Introductie	IV
1. Slimmer werken met veel data	7
2. Slim gebruik van bereiknamen	15
3. Geavanceerde formules en functies	19
4. Als Excel #WAARDE! roept – foutmeldingen ontrafeld	47
5. Formuleveiligheid en werkbladbeveiliging	55
6. Slim automatiseren met macro's	65
7. De “DB”-functies: Databases in je werkblad	69
8. Werken met meerdere gegevensbronnen	81
9. Power Query: de wasstraat voor je gegevens	93
10. Draaitabellen die je werk uit handen nemen	101
11. Power Pivot: de ruggengraat van je datamodel	109
12. Dashboards bouwen die indruk maken	117
13. Scenario's en simulaties	131
14. Excel als rapportagetool	149
15. Excel Lab	155
15.1. REGEX-functies: tekst bewerken als een taalgeleerde	155
15.2. Python in Excel: een stille revolutie	160
15.3. Bereiken combineren	168
15.4. Dynamische bereiken met # en . in Excel	172
15.5. Tekst knippen zonder gedoe	177
15.6. Pas een berekening toe op elke rij of kolom	181
15.7. Cumulatief rekenen zonder hulpkolommen	184
15.8. De Lambda-variant van een lusje	189
15.9. Keuzevakjes die echt iets doen	193
15.10. Gegevenstypen: als Excel praat met de buitenwereld	196
16. Samenwerken met Word, PowerPoint en Outlook	203
17. Excel en AI	211
18. Geografische gegevens visueel maken	215

19.	De losse eindjes.....	223
20.	Praktische projecten.....	231
20.1.	Calculator voor weerstanden.....	231
20.2.	Aandelenmonitor in Excel.....	241
20.3.	Ledenadministratie.....	248
	Appendix A: uitwerkingen praktijkopdrachten	273
	Appendix B: Index van functies.....	289
	Appendix C: Overlopende functies in Excel	291
	Bonushoofdstuk – Excel-geheimen die zelfs gevorderden vaak missen.....	295

Introductie

Ben jij al gevorderd?

Welkom terug in de wereld van Excel. Je hebt de basis inmiddels onder de knie, weet hoe je formules schrijft, gegevens sorteert en misschien zelfs hoe je een draaitabel maakt.

Maar wat als je méér wilt? Wat als je Excel wilt inzetten om je werk efficiënter te maken, betere rapporten te bouwen en zelfs kleine dashboards te maken zonder dat je meteen naar Power BI of VBA hoeft te grijpen?

Dan is dit boek er voor jou. In “Excel voor Gevorderden” gaan we voorbij aan de basis. We leren je hoe je structuur aanbrengt in je data, foutcontrole toepast, draaitabellen écht benut en krachtige functies gebruikt waar je collega’s nog nooit van gehoord hebben.

En dat over het algemeen zonder één regel programmeercode. Alleen pure Excel.

Wanneer ben je een gevorderde gebruiker?

Twijfel je of je dit boek aankunt? Hier is een korte checklist. Kun je dit met vertrouwen zeggen?

- ☐ Ik weet hoe ik absolute en relatieve celverwijzingen gebruik.
- ☐ Ik heb al eens een draaitabel aangemaakt, ook al vond ik het nog wat ingewikkeld.
- ☐ Ik weet hoe ik gegevens sorteer en filter met behulp van tabelfilters.
- ☐ Ik ben niet bang voor functies als VERT.ZOEKEN() of ALS().
- ☐ Ik heb het boek “Excel voor Beginners” gelezen.

Herken je jezelf hierin? Dan ben je klaar voor de volgende stap.

Wat kun je na dit boek?

Na het lezen van dit boek kun je onder andere:

- ✓ Grote hoeveelheden gegevens organiseren en analyseren zonder het overzicht te verliezen.
- ✓ Complexe formules schrijven die robuust en fouttolerant zijn.
- ✓ Werken met meerdere gegevensbronnen in één bestand.
- ✓ Power Query inzetten om gegevens te importeren en op te schonen.
- ✓ Visueel aantrekkelijke rapporten en dashboards maken in Excel.

Voor wie is dit boek bedoeld?

- ✓ Voor iedereen die Excel slimmer wil inzetten, van ervaren professionals tot nieuwkomers die snel impact willen maken.
- ✓ Voor zelfstandigen, herintreders én jonge data-enthousiastelingen (denk aan GenZ) die hun vaardigheden naar een hoger niveau willen tillen.
- ✓ Voor docenten, onderzoekers en projectleiders die Excel niet alleen gebruiken voor cijfers, maar voor inzichten.

Of je nu in finance, HR, marketing, onderwijs of techniek werkt, als Excel je gereedschap is, dan helpt dit boek je om het scherper te slijpen.

Hoe werkt dit boek?

Elk hoofdstuk biedt:

- ✓ Praktische uitleg met herkenbare voorbeelden.
- ✓ Tips en sneltoetsen.
- ✓ Een samenvatting.
- ✓ Reflectievragen.
- ✓ Afhankelijk van de inhoud van het hoofdstuk een praktijkopdracht.

Veel voorbeelden in dit boek gaan uit van situaties die voorkomen in de (detail)handel. Dit is echter slechts omdat deze voorbeelden zich goed lenen om uitleg te geven over toegepaste technieken en oplossingen. Maar deze technieken en toepassingen zijn uiteraard ook bruikbaar in andere omgevingen en voor andere situaties. Vele wegen leiden naar Rome; zo is het ook met Excel. De hier beschreven oplossingen en technieken zijn voorbeelden van hoe het kán, maar er zijn natuurlijk ook andere oplossingen te bedenken. Dat is het punt waarop jij als lezer in beeld komt. Laat je inspireren door dit boek, maar wees niet bang om zelf nieuwe wegen te ontdekken. Dan maak je echt gebruik van de kennis die je hier opdoet.

Dit boek is met de grootste zorgvuldigheid geschreven, maar het is mensenwerk. Alle voorbeelden zijn uitgewerkt in Excel 365 Insider met Bèta-kanaal registratie en zouden probleemloos hiermee moeten werken. Werkende bestanden¹ zijn tevens gratis te downloaden van de website <https://excelbooks7.wordpress.com>, waar ook meer informatie over Excel te vinden is, zoals de Functie van de week.

Veel plezier en succes! Reken maar dat je straks indruk maakt met je nieuwe Excelvaardigheden.

¹ Afhankelijk van de gebruikte Excel-versie.

1. Slimmer werken met veel data

Of: hoe je orde schept in de chaos zonder je hoofd te verliezen.

Zodra je Excel gaat gebruiken voor meer dan wat lijstjes of snelle berekeningen, kom je al snel op onbekend terrein: honderden of duizenden rijen aan data, met kolommen die *nét* niet altijd dezelfde betekenis lijken te hebben. Misschien importeer je gegevens uit een crm-systeem, boekhoudpakket of enquêteplatform en moet je ze zelf analyseren of opschonen.

Van rommelige data naar gestructureerde set

Grote datasets zijn alleen werkbaar als ze aan een paar fundamentele principes voldoen:

- ☐ Elk gegeven moet in zijn eigen cel staan.
- ☐ Elk kolomhoofd moet uniek en beschrijvend zijn.
- ☐ Geen lege rijen tussen de data.

Gebruik dus de importopties in Excel met beleid: liever Power Query (zie hoofdstuk 9) dan kopiëren/plakken uit een pdf. Maar als je *tóch* handmatig moet starten zorg dan dat je meteen tabellen gebruikt.

Excel-tabellen: je geheime wapen

- ☐ Zodra je een dataset in een echte Excel-tabel (**Ctrl + L**) hebt gezet, gebeurt er iets magisch:
- ☐ Je kunt er dynamisch op verwijzen met gestructureerde formules zoals `'=SOM(Tabel1[Bedrag])'`.
- ☐ Filters en sorteringen blijven actief, ook als je rijen toevoegt.
- ☐ Je formules breiden automatisch uit als je nieuwe data toevoegt.

Gegevens controleren: fouten voorkomen vóór ze ontstaan

Gebruik:

- ☐ Gegevensvalidatie om invoer te beperken.
- ☐ Voorwaardelijke opmaak om afwijkende waarden te markeren.
- ☐ Zoek dubbele gegevens om dubbele invoer te signaleren.

Navigeren en selecteren als een pro²

Actie

Naar einde kolom met gegevens
Selecteer aaneengesloten gegevens in één richting
Selecteer aaneengesloten gegevens in alle richtingen
Naar cel A1
Naar laatste cel met gegevens
In tabel hele kolom met gegevens selecteren, buiten tabel gehele kolom
In tabel hele rij met gegevens selecteren, buiten tabel gehele rij
Hele kolom selecteren, ongeacht of er gegevens staan
Hele rij selecteren, ongeacht of er gegevens staan

Toetsencombinatie

Ctrl + Pijltje
Ctrl + Shift + Pijltje
Ctrl + A
Ctrl + Home
Ctrl + End
Ctrl + Spatiebalk
Shift + Spatiebalk
Klik op kolomletter
Klik op rijnummer

Een uitstapje vóór de Praktijkopdracht

In Praktijkopdracht 1.1 wordt gebruikgemaakt van een werkblad met minimaal 500 rijen en 5 kolommen. Niet iedereen zal zoiets kant-en-klaar “op de plank” hebben liggen. Maar om 2500 cellen handmatig met willekeurige data te vullen is niet iets waar je naar uit hoeft te kijken. Maar waarom zou je Excel zelf niet zoiets laten maken? Hier komt de manier om dit werk uit handen te geven.

Wat is het probleem?

We hebben een werkblad nodig met minimaal 500 rijen en 5 kolommen aan gegevens. De gegevens moeten willekeurig zijn en we willen dit niet met de hand intypen.

Oplossing

Omdat de gegevens willekeurig moeten zijn, maken wij gebruik van de functie `ASELECT()`. De functie `ASELECT()` geeft als resultaat een gelijkmatig verdeeld willekeurig reëel getal dat groter of gelijk is aan 0 en kleiner dan 1. Telkens wanneer het werkblad wordt berekend, wordt een ander willekeurig reëel getal gegeven. Om `ASELECT()` voor ons te laten werken, gaan we het volgende doen.

- ☐ Open een nieuw Excel-bestand.
- ☐ Voor het in de Praktijkopdracht te gebruiken bestand hebben we 5 kolommen nodig:
 - Datum
 - Medewerker
 - Project

² ⚠ Let op: sneltoetsen kunnen verschillen per versie, taal- en toetsenbordinstellingen

- Uren
- Opmerking

- ☐ Zet deze kolomkoppen in rij 1 van de kolommen A tot en met E.
- ☐ Als eerste gaan we kolom A vullen met willekeurige datums, die echter wel chronologisch worden gegenereerd. Om dit te bereiken beginnen we met een startdatum, bijvoorbeeld 1 januari 2024.
- ☐ Zet in cel A2 de datum 01-01-2024. Zorg ervoor dat kolom A als opmaak “Datum” heeft.
- ☐ Zet in cel A3 de formule:

=ALS(ASELECT()>0,5;A2;INTEGER(ASELECT()*5)+A2)

- ☐ Kopieer deze formule tot en met cel A501, zodat er 500 rijen met datums ontstaan.

Hoe werkt deze formule?

De functie ASELECT() genereert een willekeurig **reëel** getal dat groter of gelijk is aan 0 en kleiner dan 1. Als ASELECT() een getal groter dan 0,5 oplevert, dan wordt de datum in de cel boven die waar de formule staat gekopieerd, zodat er ook data op dezelfde datums wordt gegenereerd. Als het getal kleiner is dan of gelijk aan 0,5, dan wordt het gegenereerde reële getal vermenigvuldigd met 5 en naar beneden afgerond op het dichtstbijzijnde gehele getal door de functie INTEGER(). Dus stel dat ASELECT() als uitkomst 0,6454 heeft, dan wordt dit vermenigvuldigd met 5=3,226999 en afgerond door INTEGER() tot 3. Dit getal wordt dan opgeteld bij de datum in de cel boven die waar de formule staat, waardoor er dus een nieuwe datum is gemaakt.

Vanaf hier zullen we de functie ASELECTTUSSEN() gebruiken, omdat deze korter is. Het is echter van belang dat het duidelijk is dat bepaalde uitkomsten op verschillende manieren te bereiken zijn.

- ☐ Zet in cel B2 de formule:

=KIEZEN(ASELECTTUSSEN(1;5); “Karel”; “Jan”; “Pieter”; “Carla”; “Anna”)

Hoe werkt deze formule?

De functie KIEZEN() gebruikt een indexgetal om een waarde uit een lijst met argumenten te geven. KIEZEN(2;"Appels";"Peren";"Kersen") geeft als resultaat "Peren", het 2^e argument in de lijst na het indexgetal. Hier wordt met behulp van ASELECTTUSSEN() een **geheel** getal van 1 tot 5 willekeurig gegenereerd en als gevolg hiervan een willekeurige naam van een medewerker in de lijst gezet.

- ☐ Kopieer deze formule tot en met cel B501, zodat er 500 rijen met namen ontstaan.
- ☐ Vervolgens is kolom C aan de beurt, waarin de naam van het project wordt weergegeven. Zet in cel C2 de volgende formule:

=KIEZEN(ASELECTTUSSEN(1;5); "Intern";"Bugfix";"Klant A";"Klant B";"Testrun")

- ☐ Kopieer deze formule tot en met cel C501, zodat er 500 rijen met projectnamen ontstaan.
- ☐ In kolom D, de kolom waarin de gewerkte uren komen, komt een iets andere formule. Omdat er hier geen teksten komen te staan, maar getallen, kan er volstaan worden met het simpelweg genereren hiervan met de volgende formule in cel D2:

=0,1+ASELECT()*12

Waarom geen ASELECTTUSSEN() in deze formule?

De functie ASELECTTUSSEN() geeft als resultaat een **geheel** getal. Wij willen echter dat ook delen van uren geregistreerd worden, dus bijvoorbeeld 8,5 uur. Vandaar dat hier ASELECT() wordt gebruikt en de uitkomst met 12 wordt vermenigvuldigd. Deze formule genereert uren van 0,1 tot en met 11,9, ruim voldoende voor ons doel.

- ☐ Kopieer deze formule tot en met cel D501, zodat er 500 rijen met gewerkte uren ontstaan.
- ☐ Maak deze cellen op als Getal en stel het aantal decimalen in op 1.

- ❑ Als laatste moet kolom E gevuld worden. Gebruik hiervoor dezelfde formule als die je gebruikt hebt bij kolom C, maar met aangepaste teksten:

=KIEZEN(ASELECTTUSSEN(1;4);"Bespreking";"Systeem A";"Systeem B";"")

- ❑ Kopieer deze formule tot en met cel E501, zodat er 500 rijen met opmerkingen ontstaan. Doordat als laatste waarde in deze formule "" is opgenomen, zal een aantal opmerkingen leeg blijven.
- ❑ Selecteer vervolgens het gehele bereik A1:D501, kopieer dit en plak het op dezelfde plaats met **Plakken Speciaal** als **Waarden**. Gebruik **Ctrl + C** om te kopiëren en dan **Ctrl + Alt + V** → Kies **Waarden** → **OK** om te plakken als vaste gegevens.

i Waarom plakken als waarden?

Door de inhoud te kopiëren en vervolgens te plakken als waarden worden de formules vervangen door de gegenereerde waarden en is er een mooie dataset ontstaan die in Praktijkopdracht 1.1 gebruikt kan worden. En dat zonder al te veel typewerk.

Als alles goed is gegaan, zien de eerste rijen van het werkblad er als volgt uit:

	A	B	C	D	E	
1	Datum	Medewerker	Project	Uren	Opmerking	
2	01/01/2024	Jan	Klant B	2,4	Bespreking	
3	04/01/2024	Pieter	Testrun	9,6		
4	05/01/2024	Pieter	Bugfix	1,9	Bespreking	
5	05/01/2024	Karel	Klant A	0,3	Systeem B	
6	05/01/2024	Jan	Testrun	0,6		
7	05/01/2024	Pieter	Bugfix	10,3	Systeem B	

De celinhoud zal natuurlijk verschillen, omdat de gegevens zijn gegenereerd met behulp van willekeurige waarden. Ook de weergave van de datum kan anders zijn: deze is afhankelijk van je instellingen in Excel.

- ❑ Selecteer vervolgens een aantal rijen, kopieer deze en plak ze onder rij 501, zodat er een paar dubbele invoeren ontstaan. Zo kun je straks testen of Excel dubbele rijen herkent met voorwaardelijke opmaak.
- ❑ Sla het bestand op met de naam UrenData.xlsx. Je gaat dit later in het boek weer gebruiken.

Nu zijn we klaar om Praktijkopdracht 1.1 uit te voeren. In een paar minuten heb je nu een dataset die realistisch oogt en perfect geschikt is om mee te oefenen. En het mooiste: je hebt meteen al vier krachtige functies leren kennen zonder één rij te hoeven typen. Win-win.



Tip:

Gebruik ASELECT.MATRIX() voor snelle testdata

Wil je supersnel een hele reeks cellen in één keer laten vullen met willekeurige getallen? Probeer dan de functie ASELECT.MATRIX() (alleen beschikbaar in Excel 365 en Excel 2019 en later).

Bijvoorbeeld:

= ASELECT.MATRIX(500;1;1;100;WAAR)

Deze formule genereert 500 willekeurige gehele getallen tussen 1 en 100 in één kolom. Ideaal als je snel testgegevens nodig hebt.



Let op: deze functie is minder geschikt als je meerdere kolommen met verschillende soorten gegevens wilt genereren (zoals in ons voorbeeld met datums, namen en uren).

Daarvoor zijn functies als ASELECT() en ASELECTTUSSEN() flexibeler in gebruik.

Praktijkopdracht 1.1 - Van chaos naar structuur

Opdrachten

- ❑ Download³ of open een werkblad met minimaal 500 rijen. (Gebruik bijvoorbeeld het bestand dat we eerder hebben gemaakt.)

³ Op excelbooks7.wordpress.com is een geschikt bestand te downloaden

- ☐ Zet deze om in een gestructureerde Excel-tabel.
- ☐ Pas validatie toe op ten minste 2 kolommen.
- ☐ Markeer dubbele invoer en afwijkende waarden.
- ☐ Gebruik Ctrl-toetscombinaties om efficiënt te navigeren.

Ben je benieuwd hoe je dit soort data in een draaitabel kunt samenvatten? In hoofdstuk 10 leer je hoe je draaitabellen maakt die je werk uit handen nemen.

Uitwerking van Praktijkopdracht 1.1

Voorbeeld: dataset met gewerkte uren (Datum, Medewerker, Project, Uren, Opmerking).

- ☐ Zet de data om naar een tabel, noem deze `UrenData`. (Gebruik de toetsencombinatie **Ctrl + L**.)
- ☐ Voeg **validatie** toe aan de kolom Uren (bijv. 0–12) en aan bijvoorbeeld de kolom Medewerker. (**Gegevens** → **Gegevensvalidatie**.)
- ☐ Gebruik **voorwaardelijke opmaak** om dubbele rijen te signaleren. (**Start** → **Voorwaardelijke Opmaak**.)
- ☐ Gebruik **sneltoetsen** om grote delen te navigeren.

Als alles goed is gegaan, moet je tabel er ongeveer als volgt uitzien, met rij 2 gemarkeerd omdat er verderop in de tabel een rij bestaat met exact dezelfde data. In rij 5 is de lege cel in de kolom 'Opmerking' gemarkeerd.

	A	B	C	D	E
1	Datum	Medewerker	Project	Uren	Opmerking
2	26-03-2024	Fatima	Marketing	3,1	Beheer
3	08-02-2024	Joost	Marketing	4,8	Bugfix
4	09-02-2024	Fatima	Interne IT	7,1	Overleg
5	25-03-2024	Joost	Interne IT	7,5	
6	03-02-2024	Alex	Marketing	6,1	Bugfix
7	27-01-2024	Alex	Klant A	4,3	Bugfix

☒ De volledige stap-voor-stapuitwerking van deze opdracht vind je op pagina 273.

Reflectievragen:

1. Welke kolommen in jouw dataset gaven de meeste problemen of vergden de meeste correcties?
 2. Welke invoerregels zou jij in je eigen werk willen instellen om fouten te voorkomen? Noem er minstens twee.
 3. Wanneer zou jij twee rijen als écht dubbel beschouwen? Welke criteria gebruik je daarbij?
 4. Aan welke signalen herken jij invoerfouten in jouw eigen werkbladen? Bedenk concrete voorbeelden.
-

Samenvatting Hoofdstuk 1:

- Structuur is alles: tabellen, kopteksten, geen lege rijen.
 - Validatie voorkomt fouten.
 - Sneltoetsen maken grote hoeveelheden gegevens beheersbaar.
 - Goede voorbereiding bespaart tijd bij analyse en rapportage.
-

2. Slim gebruik van bereiknamen

Of: hoe je formules schrijft die zelfs een collega snapt.

Je hebt het vast wel eens meegemaakt: je opent een werkblad van een collega, ziet de formule =SOM(B2:B45) en denkt: wat was dit ook alweer? Of je komt =B2*C2 tegen en hebt geen idee waar die getallen voor staan. Excel rekent het allemaal wel netjes uit, maar voor mensen zijn die formules vaak abracadabra.

Daarom zijn bereiknamen zo krachtig. Ze maken je werkbladen leesbaarder, onderhoudsvriendelijker en vooral: minder foutgevoelig.

Wat is een bereiknaam?

Een bereiknaam is een zelfgekozen naam voor een cel of een groep cellen. In plaats van =SOM(B2:B45) kun je schrijven:

=SOM(OmzetPerMaand)

Dat leest een stuk prettiger, zeker als je maanden later weer in het bestand duikt (of iemand anders dat doet).

Waarom bereiknamen gebruiken?

- ☐ Betere leesbaarheid – formules zijn begrijpelijker: =PrijsExcl * BTW is duidelijker dan =B2*C2.
- ☐ Minder fouten – als je bereik verandert, hoef je maar op één plek iets aan te passen.
- ☐ Makkelijker doorverwijzen – ook tussen werkbladen of zelfs werkmappen.
- ☐ Handig bij gegevensvalidatie, keuzelijsten en dynamische formules.

Hoe geef je een bereik een naam?

Er zijn verschillende manieren:

Via het **naamvak** (snel):

- ☐ Selecteer de cel of het bereik
- ☐ Klik in het naamvak linksboven (waar normaal “B2” of zo iets staat).
- ☐ Typ de gewenste naam, bijvoorbeeld OmzetPerMaand en druk op **Enter**.

Via **Formules** → **Namen definiëren**:

- ☐ Ga naar het tabblad Formules.
- ☐ Kies Namen beheren of Naam definiëren.

- ☐ Hier kun je ook bereiken een beschrijving geven of dynamisch maken (zie verderop).

Regels voor namen

Er zijn een paar regels waar bereiknamen aan moeten voldoen:

- ✓ Begin met een letter (dus niet met een getal).
- ✓ Geen spaties: gebruik bijvoorbeeld Omzet_Jan, TotaalKosten, BTWpercentage.
- ✓ Hoofdletters en kleine letters maken geen verschil.

Voorbeeld:

Btw-berekening met bereiknamen.

Stel, je hebt:

In cel A1: prijs exclusief btw (bijv. 100)

In cel B1: btw-percentage (bijv. 21%)

- ☐ Geef A1 de naam PrijsExcl en B1 de naam BTW.
- ☐ De formule in C1 wordt dan:

=PrijsExcl * (1 + BTW)

Veel duidelijker dan =A1*(1+B1)!

Tip:

Bereiknamen gebruiken bij gegevensvalidatie.

Stel je hebt een lijst met categorieën in cellen A1:A5, bijvoorbeeld:

- Inkoop
- Verkoop
- Marketing
- Onderzoek
- Diversen

Geef dit bereik de naam Categorieën. Bij gegevensvalidatie kun je dan bij “Bron” invullen:

=Categorieën

Zo koppel je eenvoudig een keuzelijst aan een bereik.

Dynamische bereiknamen (geavanceerd)

Je kunt een bereiknaam zo instellen dat deze automatisch meegroeit als je nieuwe data toevoegt. Dit doe je via het **tabblad Formules**:

- ☐ Ga naar **Formules** → **Naam definiëren**.
- ☐ Geef het bereik een naam, bijvoorbeeld DynamischBereik.
- ☐ In het veld Verwijst naar zet je een formule zoals:

=VERSCHUIVING(Blad1!\$A\$1;0;0;AANTALARG(Blad1!\$A:\$A);1)

Deze formule maakt een bereik vanaf A1 dat automatisch net zoveel rijen omvat als er gegevens in kolom A staan.

⚠ Let op: deze formule **werkt niet** in het **naamvak** linksboven, maar alleen via **Naam definiëren** in het lint.

Een eenvoudiger alternatief: zet je data in een tabel (**Ctrl + L**). Dan kun je het bereik benoemen op basis van de kolomtitel:

=Tabel1[Bedrag]

Praktijkopdracht 2.1 - Slim rekenen met bereiknamen

Situatie: je beheert een Excel-bestand met bestellingen van producten, inclusief aantallen, prijs exclusief btw en btw-tarief. Je wilt een overzicht dat automatisch meerekent én begrijpelijk blijft voor collega's.

Opdrachten:

- ☐ Maak een werkblad met de kolommen: Product, Aantal, PrijsExcl, BTW%.
- ☐ Voeg de kolom TotaalIncl toe, met de formule om de prijs inclusief btw te berekenen.
- ☐ Geef de kolommen Aantal, PrijsExcl, BTW%, TotaalIncl elk een bereiknaam.
- ☐ Gebruik deze bereiknamen in je formule voor TotaalIncl.
- ☐ Maak op een apart werkblad een samenvatting per btw-tarief met het totaalbedrag (hint: gebruik SOM.ALS() of SOMMEN.ALS() in combinatie met bereiknamen).
- ☐ Experimenteer met het aanpassen van invoer (zoals aantallen of prijzen) en observeer hoe de berekeningen automatisch mee veranderen.
- ☐ Controleer of de formules goed leesbaar zijn voor iemand anders dan jijzelf. Pas namen eventueel aan voor extra duidelijkheid.

Uitwerking van Praktijkopdracht 2.1

- ☐ Maak een werkblad aan met de kolommen Product, Aantal, PrijsExcl en BTW%.
- ☐ Gebruik het tabblad **Formules** → **Naam definiëren** om in de kolommen Aantal, PrijsExcl, BTW en TotaalIncl de eerste 20 rijen als bereiknaam vast te leggen.
- ☐ Gebruik deze namen in je berekening.
- ☐ Voeg een kolom toe met de naam TotaalIncl waarin je de volgende formule opneemt:

=Aantal * PrijsExcl * (1 + BTW)

- ☐ Zet op een apart werkblad in cel A1 een btw-tarief, bijvoorbeeld 21%.
- ☐ Zet in cel A2 een tweede, ander btw-tarief, bijvoorbeeld 9%.
- ☐ Geef deze twee cellen elk een naam, bijvoorbeeld “hoog” en “laag”.
- ☐ Bereken het totaalbedrag per btw-tarief door gebruik te maken van de functies SOM.ALS() of SOMMEN.ALS() in combinatie met bereiknamen.

Een werkend Excel-bestand is beschikbaar voor download op <https://excelbooks7.wordpress.com>.

Reflectievragen:

1. Wat maakt de formule met bereiknamen in jouw opdracht duidelijker dan een formule met celverwijzingen?
2. Welke namen helpen jou het meest om overzicht te houden in grotere werkbladen? Bedenk voorbeelden.
3. Hoe gebruik je bereiknamen in werkmappen met meerdere tabbladen zonder verwarring te veroorzaken?

Samenvatting Hoofdstuk 2:

- Bereiknamen maken je formules leesbaar en begrijpelijk.
 - Je kunt ze eenvoudig instellen via het naamvak of via het tabblad Formules.
 - Ze zijn ideaal voor gegevensvalidatie, berekeningen en dynamische werkbladen.
 - Kies duidelijke namen zonder spaties.
 - Gebruik bij voorkeur Excel-tabellen om dynamische namen nog krachtiger te maken.
-

3. Geavanceerde formules en functies

Of: hoe je met één formule ineens vijf kolommen onder controle hebt.

In dit hoofdstuk duiken we dieper in een aantal krachtige functies die Excel veel flexibeler en slimmer maken. We laten je zien hoe je logica, tekstbewerking, dynamische arrays en zelfgemaakte functies inzet om werkbladen écht te laten doen wat jij wilt. Bij elke functie krijg je de syntaxis, een kort voorbeeld, een uitgewerkte toepassing én extra uitleg voor inzicht. Geen lange theorie – gewoon dingen die werken.

Overlopende functies

Wat zijn overlopende functies? Sommige functies in Excel geven niet één uitkomst, maar een hele reeks waarden, bijvoorbeeld een rij klanten, een lijst unieke producten of een reeks getallen. Deze functies lopen over: ze zetten hun resultaten automatisch in meerdere cellen, meestal onder of naast de formulecel.

Dit type functies noemen we:

Overlopende matrixfuncties

(in het Engels: spill functions)

Je hoeft ze niet naar beneden te kopiëren; Excel zorgt zelf voor de juiste ruimte. Maar: als er cellen in de weg staan, krijg je de foutmelding #OVERLOOP!.

Voorbeelden van overlopende functies:

FILTER() – geeft alle rijen die aan een voorwaarde voldoen.

UNIEK() – toont alleen de unieke waarden uit een bereik.

REEKS() – maakt een reeks getallen in één keer.

X.ZOEKEN() – kan meerdere resultaten tegelijk teruggeven.

TEKST.SPLITSEN() – splitst een tekst in meerdere kolommen of rijen.

 **Tip:** Zorg altijd voor voldoende lege ruimte onder of naast de formulecel en wees alert bij het kopiëren of verplaatsen van overlopende formules: je hebt te maken met meerdere cellen tegelijk, ook al lijkt het op het eerste gezicht maar één formule.

Zie ook: Appendix C: Lijst van overlopende functies.

3.1. Geneste logische functies

3.1.1. ALS()

Gebruik ALS() om op basis van een voorwaarde uit twee uitkomsten te kiezen.

Syntaxis:

=ALS(logische_test;waarde_als_waar;[waarde_als_onwaar])

Voorbeeld:

=ALS(B2>100;"Hoog";"Laag")

Stel je hebt een beoordelingsformulier voor cursisten. In kolom B staan eindcijfers en je wilt automatisch een beoordeling tonen in de vorm van letters (A t/m D).

Hiervoor kun je geneste ALS()-functies gebruiken:

=ALS(B2>=90;"A";ALS(B2>=80;"B";ALS(B2>=70;"C";"D")))

i Wat gebeurt hier?

Als B2>=90 WAAR is, dan wordt de letter "A" teruggegeven; anders wordt de volgende ALS()-functie uitgevoerd;

Als B2>=80 WAAR is, dan wordt de letter "B" teruggegeven; anders wordt de volgende ALS()-functie uitgevoerd;

Als B2>=70 WAAR is, dan wordt de letter "C" teruggegeven, maar als deze test ONWAAR is, dan wordt de letter "D" teruggegeven.

Zodra een logische test de waarde WAAR oplevert, wordt de formule niet verder uitgevoerd en de bijbehorende waarde teruggegeven.

3.1.2. ALS.VOORWAARDEN()

Syntaxis:

=ALS.VOORWAARDEN(test1;resultaat1;[test2;resultaat2];...)

Voorbeeld:

=ALS.VOORWAARDEN(B2>=90;"A";B2>=80;"B";B2>=70;"C";WAAR;"D")

Het voorbeeld hierboven met de geneste ALS()-functies kun je overzichtelijker maken door ALS.VOORWAARDEN() te gebruiken. Hiermee krijg je verschillende uitkomsten, afhankelijk van aan welke voorwaarde wordt voldaan.

Let op de waarde “WAAR” die als het één na laatste argument is opgenomen. Dat is de “**vangnet**”-voorwaarde, voor alle gevallen die niet aan de vorige voorwaarden voldoen.

Wat doet WAAR hier precies?

WAAR is altijd waar. Dus:

Als geen van de andere voorwaarden geldt, dan valt Excel automatisch terug op de regel met WAAR.

Het is hetzelfde idee als de “anders” in een klassieke ALS()-functie.

WAAR is de “**als niets anders klopt**”-optie.

Het is verplicht om *iets* op te geven voor het geval géén van de andere voorwaarden geldt - anders krijg je een foutmelding. Door WAAR als voorwaarde te gebruiken, zorg je ervoor dat Excel altijd iets kan teruggeven.

3.2. ASELECT.MATRIX() – willekeurige data genereren

Handig als je snel testgegevens nodig hebt, bijvoorbeeld bij het bouwen van een dashboard.

Syntaxis:

=ASELECT.MATRIX([rijen];[kolommen];[minimum];[maximum]; [geheel_getal])

Voorbeeld:

=ASELECT.MATRIX(10;5;1;100;WAAR)

Met deze argumenten genereert deze functie 10 willekeurige, gehele getallen tussen 1 en 100 (beide inclusief) in 5 kolommen met 10 rijen. Als WAAR vervangen wordt door ONWAAR, dan worden 10 willekeurige reële getallen teruggegeven.

 **Let op:** in Excel wordt bij deze functie gesproken over een “decimaal getal”, maar eigenlijk bedoelt men: een niet-geheel getal, dus een getal met cijfers achter de komma. “Decimaal getal” betekent letterlijk: een getal in het tientallig stelsel. Dat zijn dus eigenlijk alle getallen die we dagelijks gebruiken. “kommagetal” of “reëel getal” zou technisch correct en een betere term zijn.

3.3. SOM() als logische EN/OF

Stel, je hebt in kolom A een twintigtal rijen met prijzen staan en in kolom B eenzelfde aantal rijen met producten, bijvoorbeeld laptops, monitoren en muizen. Als je nu wilt weten hoeveel laptops een prijs van meer dan € 1000 hebben, dan kun je gebruikmaken van een functie waarvan je niet verwacht dat deze zich daarvoor leent: SOM(). Je kunt zelfs een EN- of een OF-test doen.

Voorbeeld met EN:

=SOM((A2:A21>1000)*(B2:B21="Laptop"))

Telt het aantal rijen waarin de prijs groter dan 1000 is én het product "Laptop". Door de functie SOMPRODUCT() te gebruiken, kun je het totaalbedrag bepalen van laptops die meer dan 1000 kosten.

Voorbeeld van som van bedragen bij beide voorwaarden (EN):

=SOMPRODUCT((A2:A21>1000)*(B2:B21="Laptop")*A2:A21)

Voorbeeld met OF:

=SOM((A2:A21>1000)+(B2:B21="Laptop"))

Telt het aantal rijen waarin de prijs groter dan 1000 is óf het product "Laptop". Als in een rij de prijs groter dan 1000 is én het product "Laptop", dan wordt deze rij dubbel geteld.

 **Let op:** de functie SOMPRODUCT() kan niet gebruikt worden in combinatie met een OF-test.

3.4. FILTER() – Dynamisch filteren met meerdere voorwaarden

De functie FILTER() is een overlopende matrixfunctie. Ze toont automatisch alle rijen die voldoen aan de opgegeven voorwaarde(n) en zet deze onder elkaar. De resultaten lopen over in de cellen eronder, zonder dat je hoeft te slepen of te kopiëren.

Syntaxis:

=FILTER(bereik;criteria;[als_leeg])

Argument	Betekenis
bereik	Het bereik dat je wilt filteren.
criteria	De voorwaarde(n) waar de rijen aan moeten voldoen.
[als_leeg]	(Optioneel) Wat er getoond wordt als er géén rijen voldoen

Voorbeeld:

=FILTER(A2:D500;(C2:C500="Zuid")*(D2:D500>1000))

klanten uit 'Zuid' met omzet > 1000

Stel je hebt een tabel in bereik A2:D500 met klanten, producten, de regio waar zij wonen en het bedrag dat zij per product hebben besteed.

Je wilt nu alleen de klanten zien die uit regio Zuid afkomstig zijn met een omzet groter dan € 1000.

Hiervoor kun je de FILTER()-functie gebruiken:

=FILTER(A2:D500;(C2:C500="Zuid")*(D2:D500>1000))

i Wat gebeurt hier?

C2:C500="Zuid" controleert de regio per rij.

D2:D500>1000 controleert of de omzet boven de 1000 is.

Door de twee voorwaarden met * te combineren, zeg je eigenlijk: "alleen als allebei waar zijn" (EN-logica).

FILTER() geeft alle rijen terug waarvoor dit geldt.

De uitkomst verschijnt direct onder je formule, verdeeld over meerdere cellen (overlopende matrixfunctie).


💡 Tip: Gebruik de derde parameter om een melding te tonen als er géén resultaten zijn:

=FILTER(A2:D500;(C2:C500="Zuid")*(D2:D500>1000); "Geen resultaten").

Uiteraard staat het je vrij om de argumenten "Zuid" en ">1000" te vervangen door een verwijzing naar een cel waar je deze waarden kunt invoeren, waardoor het

Excel voor Gevorderden

gebruik van de functie flexibeler wordt. Als je dit doet én het argument [als_leeg] gebruikt, dan zou je zoiets als dit op je scherm moeten zien:



The screenshot shows an Excel spreadsheet with the following data and formula:

	A	B	C	D	E	F	G	H	I	J	K
1	Klant	Product	Regio	Omzet							
2	Jansen	Printer	Zuid	€ 1.200,00		Regio	Zuid	Jansen	Printer	Zuid	€ 1.200,00
3	De Vries	Scanner	Noord	€ 900,00		Omzet	1000	Willems	Laptop	Zuid	€ 2.500,00
4	Willems	Laptop	Zuid	€ 2.500,00							
5	Peters	Muis	Zuid	€ 35,00							

The formula bar shows: `=FILTER(A2:D500;(C2:C500=G2)*(D2:D500>G3);"Geen resultaten")`

De formule luidt dan:

`=FILTER(A2:D500;(C2:C500=G2)*(D2:D500>G3);"Geen resultaten")`

Opmerking: als de functie meerdere waarden teruggeeft en overloopt, verschijnt er een dun blauw kader rond het hele resultaatgebied. Dit helpt je te zien welke cellen bij het overlopende resultaat horen. Daarnaast toont Excel ook visuele aanwijzingen zoals het ontbreken van een formule in de overloopcellen zelf.

3.5. LET() en LAMBDA() – formulemodulariteit

3.5.1. LET()

helpt je om namen toe te wijzen aan berekeningen binnen één formule. Zo voorkom je dubbele berekeningen en maak je de formule beter leesbaar.

Syntaxis:

`=LET(naam1, waarde1, berekening_of_naam2, [waarde2, berekening_of_naam3...])`

Je definieert namen voor onderdelen van je berekening en gebruikt die vervolgens in je uiteindelijke formule.

Argument

naam1

waarde1

berekening_of_naam2

waarde2

Betekenis

De naam die je zelf kiest voor een waarde in de formule. Moet beginnen met een letter.

De waarde die is toegewezen aan naam1.

Een berekening die gebruikmaakt van alle namen in de LET-functie. Dit moet het laatste argument zijn in de LET-functie, óf

Een tweede naam die wordt toegewezen aan een tweede waarde. Als een naam is opgegeven, worden waarde2 en berekening_of_naam3 vereist.

(Optioneel) De waarde die is toegewezen aan berekening_of_naam2.

berekening_of_naam3 (Optioneel) Een berekening die gebruikmaakt van alle namen in de LET-functie. Het laatste argument in de LET-functie moet een berekening zijn, óf Een derde naam die wordt toegewezen aan een derde waarde. Als een naam is opgegeven, worden waarde3 en berekening_of_naam4 vereist.

Stel, je hebt het volgende werkblad en je wilt op basis van de verkoopprijs je marge berekenen.

	A	B	C	D
1	Product	Kosten	Prijs	Marge
2	Lamp	€ 87,00	€ 113,00	23%
3	Stoel	€ 93,00	€ 150,00	38%

In cel D2 zou je een formule kunnen zetten die elke keer opnieuw de berekening maakt, maar met LET() kun je dit als volgt oplossen door dit in cel D2 te zetten:

=LET(Kosten;B2;Prijs;C2;Marge;(Prijs-Kosten)/Prijs;Marge)

i Wat gebeurt hier?

Je definieert:

Kosten = B2

Prijs = C2

Marge = (Prijs - Kosten)/Prijs

En **het laatste argument** (Marge) is de “output” van de formule: datgene wat Excel uiteindelijk laat zien in de cel.

Je kunt deze formule in D2 gewoon naar beneden kopiëren en zo voor elke rij de marge berekenen.

3.5.2. LAMBDA()

stelt je in staat om je eigen functies te definiëren.

Syntaxis:

=LAMBDA([parameter1, parameter2, ...,] berekening)

Excel voor Gevorderden

Je definieert invoerwaarden (parameters) en schrijft daarna de berekening waarin je die gebruikt.

Argument	Betekenis
parameter, parameter	(Optioneel) Een waarde die je aan de functie wilt doorgeven, zoals een celverwijzing, tekenreeks of getal. Je kunt maximaal 253 parameters invoeren.
berekening	De formule die je wilt uitvoeren en retourneren als resultaat van de functie. Het moet het laatste argument zijn en moet een resultaat retourneren.

Stel je hebt het eerder gebruikte werkblad uitgebreid met een kolom Incl.BTW en je wilt hier de prijs inclusief BTW berekenen.

	A	B	C	D	E
1	Product	Kosten	Prijs	Marge	Incl. BTW
2	Lamp	€ 87,00	€ 113,00	23%	€ 136,73
3	Stoel	€ 93,00	€ 150,00	38%	€ 181,50

In cel E2 kun je in combinatie met LET() een krachtige, herbruikbare formule maken:

=LET(Factor;1,21;BTWcalc;LAMBDA(x; x*Factor);BTWcalc(C2))

Wat gebeurt hier?


Je definieert:

Factor = 1,21	Je definieert het btw-percentage (21%) als vaste waarde
BTWcalc = ...	Je maakt een functie genaamd BTWcalc
x =	De parameter van de functie (de invoerwaarde)
x * Factor =	de berekening die op x moet worden uitgevoerd
BTWcalc(C2) =	Je voert de functie uit met de waarde in cel C2

In het hierboven beschreven voorbeeld maakt de functie LAMBDA() deel uit van een formule. Een andere, eigenlijk nóg krachtigere, toepassing van LAMBDA() is echter het creëren van je eigen functies, die je steeds opnieuw, net als een “echte” functie in Excel, kunt herbruiken. Laten wij eens kijken hoe dit werkt.

Een mooi voorbeeld waar zo'n zelfgemaakte functie van pas kan komen is bij het gebruik van de stelling van Pythagoras, een wiskundige stelling die het verband geeft tussen de lengten van de zijden van een rechthoekige driehoek. Hoewel de berekening niet ingewikkeld is, is het wel een goed voorbeeld om de kracht van LAMBDA() te demonstreren.

De (welbekende) stelling luidt: "In een rechthoekige driehoek is de som van de kwadraten van de lengtes van de rechthoekszijden gelijk aan het kwadraat van de lengte van de schuine zijde."



Een mooi opstapje om met LAMBDA() aan de gang te gaan. De uitgeschreven wiskundige vorm is (eveneens welbekend):

$$a^2 + b^2 = c^2$$

Om voor de stelling een functie te maken, gaan we als volgt te werk:

Geef de functie een naam via **Formules** → **Gedefinieerde namen** → **Naam definiëren**.

Maak de functie beschikbaar in de gehele werkmap.

Gebruik als bron de formule:

=LAMBDA(a;b;WORTELT(a^2+b^2))

Als de functie de naam "Pyth" heeft gekregen, dan kun je deze nu binnen de hele werkmap, bijvoorbeeld als volgt, gebruiken:

=Pyth(A2;B2)

als in cel A2 de lengte van zijde a staat en in cel B2 de lengte van zijde b.

Als je dan in cel C2 je functie zet, dan krijg je, als alles goed is gegaan, zoiets te zien: